SU3-ASYM

A companion *Mathematica* package to the paper "On the number of n-dimensional representations of $SU(3)$, the Bernoulli numbers, and the Witten zeta function"

Dan Romik
Department of Mathematics
University of California, Davis

October 4, 2016

This notebook was created using *Mathematica* 8 and later updated (version 1.2) using *Mathematica* 11.

Release notes:

- **Version 1.0** (March 11, 2015). Initial version.
- **Version 1.1** (March 12, 2015). Added section 5 with code for numerical computation of the sequence r(n).
- **Version 1.2** (October 4, 2016). Updated equation numbers and the formulas for the constant K to correspond to the October 1, 2016 revision of the paper.

1. Introduction

This *Mathematica* package accompanies my paper "On the number of n-dimensional representations of $SU(3)$, the Bernoulli numbers, and the Witten zeta function". Its goal is to help readers of the paper quickly perform algebraic and numerical computations that verify a few claims made in the paper.

The algebraic computations can all be done by hand in a few minutes, but the computer-aided verification is a lot easier and less time-consuming.

To proceed, open each of the sections below and follow the instructions.

2. Bernoulli numbers

Verification of Lemma 2.2

This section verifies equation (33) in the paper used in the proof of Lemma 2.2. First, run the code below to define the functions F[n,k], R[n,k] and G[n,k], and the expressions on the two sides of equation (33), denoted LHS33 and RHS33. The code also defines a useful function called SimplifyTogether[] that brings calculation results to a pleasant form.

In[1]:=

```
SimplifyTogether[expr_] := Together[FullSimplify[expr]];

F[n_, k_] := 1/(3 n + 1) × (2 n + 1) !/(n !) ^2 × Binomial[n, k]/Binomial[3 n, n + k];

R[n_, k_] := k (2 n - k + 1) (11 n^2 + 2 k^2 - 5 n k + 27 n - 6 k + 16)/(3 (n + 1) (3 n + 2) (3 n + 4) (n - k + 1));

G[n_, k_] := F[n, k] R[n, k];

LHS33 = F[n + 1, k] - F[n, k];
RHS33 = G[n, k] - G[n, k + 1];
```

The next line will show what LHS33/F[n,k] simplifies to after applying the SimplifyTogether function. Run it to see what happens.

In[7]:=

```
SimplifyTogether[LHS33 / F[n, k]]
```

Out[7]=

$$\left(12 - 18 k + 12 k^2 - 6 k^3 + 46 n - 68 k n + 26 k^2 n - 4 k^3 n + 67 n^2 - 77 k n^2 + 12 k^2 n^2 + 44 n^3 - 27 k n^3 + 11 n^4\right) / (3 (-1 + k - n) (1 + n) (2 + 3 n) (4 + 3 n))$$

Now do the same for RHS33/F[n,k].

In[8]:=

```
SimplifyTogether[RHS33 / F[n, k]]
```

Out[8]=

$$\left(-12 + 18 k - 12 k^2 + 6 k^3 - 46 n + 68 k n - 26 k^2 n + 4 k^3 n - 67 n^2 + 77 k n^2 - 12 k^2 n^2 - 44 n^3 + 27 k n^3 - 11 n^4\right) / (3 (1 + n) (1 - k + n) (2 + 3 n) (4 + 3 n))$$

You can probably see that the two outputs are the same, but just to make sure, run SimplifyTogether on the difference of LHS33 and RHS33 divided by F[n,k]. If you get 0, that means equation (33) is correct.

In[9]:=

```
SimplifyTogether[(LHS33 - RHS33) / F[n, k]]
```

Out[9]=

```
0
```

Verification of Lemma 2.3

Here, we verify equation (38) used in the proof of Lemma 2.3. First, run the code below to load the definitions of the functions F1, F2, U, V, W, R1, R2, G1, G2.

In[10]:=
```
SimplifyTogether[expr_] := Together[FullSimplify[expr]];
(* Repeat the definition of SimplifyTogether from the
 previous section so both sections can be run independently *)

F1[n_, m_, j_] := (-1)^(n+j) Binomial[n, j] Binomial[3 n, m + j - n] / Binomial[3 n, n + j];
F2[n_, m_, j_] := Binomial[n, j] Binomial[n + j - 1, m + j - n] / Binomial[3 n, n + j];

U[n_, m_] := (m + 2) (2 n - m - 1);
V[n_, m_] := -2 m^2 - 5 n^2 + 8 m n + 9 n - 4 m - 2;
W[n_, m_] := - (4 n - m) (2 n - m - 1);

R1[n_, m_, j_] := j (3 n + 1) (2 n - j + 1) / (n - m - j - 1);
R2[n_, m_, j_] := j (m - 2 n + 1) (2 n - j + 1) / (n - m - j - 1);

G1[n_, m_, j_] := F1[n, m, j] R1[n, m, j];
G2[n_, m_, j_] := F2[n, m, j] R2[n, m, j];
```

Now, define the expressions on the left-hand side and the right-hand side of equation (38). Note that there is a parameter α that takes the values α=1 and α=2, so we define two quantities, LHS38one and LHS38two, for the left-hand side (corresponding to the two values of α), and similarly for the right-hand side.

In[20]:=
```
LHS38one = U[n, m] F1[n, m + 2, j] + V[n, m] F1[n, m + 1, j] + W[n, m] F1[n, m, j];
RHS38one = G1[n, m, j + 1] - G1[n, m, j];

LHS38two = U[n, m] F2[n, m + 2, j] + V[n, m] F2[n, m + 1, j] + W[n, m] F2[n, m, j];
RHS38two = G2[n, m, j + 1] - G2[n, m, j];
```

The next four lines of code show what happens when we SimplifyTogether each of the expressions LHS38one/F1[n,m,j], RHS38one/F1[n,m,j], LHS38two/F2[n,m,j], RHS38two/F2[n,m,j].

In[24]:=
```
SimplifyTogether[LHS38one / F1[n, m, j]]
```

Out[24]=
$$-\frac{(1+3n)\left(-2j-2jm+2jn+j^2n+mn-2jmn-4n^2+3jn^2+mn^2-4n^3\right)}{(1+j+m-n)(2+j+m-n)}$$

In[25]:=
```
SimplifyTogether[RHS38one / F1[n, m, j]]
```

Out[25]=
$$-\frac{(1+3n)\left(-2j-2jm+2jn+j^2n+mn-2jmn-4n^2+3jn^2+mn^2-4n^3\right)}{(1+j+m-n)(2+j+m-n)}$$

In[26]:=
```
SimplifyTogether[LHS38two / F2[n, m, j]]
```

Out[26]=
$$\frac{(1+m-2n)\left(2j+jm-j^2m+3jn+4j^2n+2jmn-n^2-3jn^2-n^3\right)}{(1+j+m-n)(2+j+m-n)}$$

In[27]:=
```
SimplifyTogether[RHS38two / F2[n, m, j]]
```

Out[27]=
$$\frac{(1+m-2n)\left(2j+jm-j^2m+3jn+4j^2n+2jmn-n^2-3jn^2-n^3\right)}{(1+j+m-n)(2+j+m-n)}$$

Again, we can visually see that LHS38one/F1[n,m,j] == RHS38one/F1[n,m,j] and that LHS38two/F2[n,m,j] == RHS38two/F2[n,m,j]. To verify this, SimplifyTogether the differences; if the answers are 0, that means equation (34) is correct.

In[28]:=
```
SimplifyTogether[(LHS38one - RHS38one) / F1[n, m, j]]
```

Out[28]=
```
0
```

In[29]:=
```
SimplifyTogether[(LHS38two - RHS38two) / F2[n, m, j]]
```

Out[29]=
```
0
```

New Bernoulli number summation identities

This subsection numerically verifies for small values of *n* the two new Bernoulli number summation identities that appear in Section 12.1 of the paper.

First, we define the functions P[n,k] and Q[n,k], and the expressions on the left-hand side and right-hand side of equations (107) and (108).

In[30]:=

```
P[n_, k_] := (2 n - 1)^2 - 4 (k - 1) (n - k);
Q[n_, k_] := ((k^2 - k n + n^2) (-1 + 4 k^2 - 4 k n + 4 n^2) - 6 k (k - n) n) / (n (2 k + 1) (2 n - 2 k + 1));

LHS107 = BernoulliB[6 n - 2] / (6 n - 2);
RHS107 = -1/(2 (6 n - 1)) Binomial[4 n, 2 n] Sum[Binomial[2 n, 2 k - 1]
      P[n, k] BernoulliB[2 n + 2 k - 2]/(2 n + 2 k - 2) BernoulliB[4 n - 2 k]/(4 n - 2 k), {k, 1, n}];

LHS108 = BernoulliB[6 n] / (6 n);
RHS108 = -2/(3 (6 n + 1)) (4 n + 1)!/((2 n)!)^2 Sum[
     Binomial[2 n, 2 k] Q[n, k] BernoulliB[2 n + 2 k]/(2 n + 2 k) BernoulliB[4 n - 2 k]/(4 n - 2 k), {k, 0, n}];
```

The next two lines of code illustrate the identity in equation (107). Run them and see what happens.

In[36]:=

```
Table[LHS107, {n, 1, 7}]
```

Out[36]=

$$\left\{-\frac{1}{120}, \frac{1}{132}, -\frac{3617}{8160}, \frac{77\,683}{276}, -\frac{3\,392\,780\,147}{3480}, \frac{151\,628\,697\,551}{12}, -\frac{261\,082\,718\,496\,449\,122\,051}{541\,200}\right\}$$

In[37]:=

```
Table[RHS107, {n, 1, 7}]
```

Out[37]=

$$\left\{-\frac{1}{120}, \frac{1}{132}, -\frac{3617}{8160}, \frac{77\,683}{276}, -\frac{3\,392\,780\,147}{3480}, \frac{151\,628\,697\,551}{12}, -\frac{261\,082\,718\,496\,449\,122\,051}{541\,200}\right\}$$

The next two lines similarly illustrate the identity in equation (108).

In[38]:=

```
Table[LHS108, {n, 1, 7}]
```

Out[38]=

$$\left\{\frac{1}{252}, -\frac{691}{32\,760}, \frac{43\,867}{14\,364}, -\frac{236\,364\,091}{65\,520}, \frac{1\,723\,168\,255\,201}{85\,932}, -\frac{26\,315\,271\,553\,053\,477\,373}{69\,090\,840}, \frac{1\,520\,097\,643\,918\,070\,802\,691}{75\,852}\right\}$$

In[39]:=
```
Table[RHS108, {n, 1, 7}]
```

Out[39]=
$$\left\{\frac{1}{252}, -\frac{691}{32\,760}, \frac{43\,867}{14\,364}, -\frac{236\,364\,091}{65\,520}, \frac{1\,723\,168\,255\,201}{85\,932}, -\frac{26\,315\,271\,553\,053\,477\,373}{69\,090\,840}, \frac{1\,520\,097\,643\,918\,070\,802\,691}{75\,852}\right\}$$

3. Numerical values of constants

In this section we define the numerical constants X, Y, A_1, A_2, A_3, A_4, K that appear in the paper, and compute their numerical values. We also compute the derivative $\omega'(0)$ of the Witten zeta function of SU(3), and the “intriguing integral” that appears in the formula for $\omega'(0)$.

The constant K is defined in terms of two different formulas, so we also check that they are equivalent.

Run the code below to load the definitions. Note that the evaluation will take a few seconds. This time is spent doing the high-precision numerical integration to compute “IntriguingIntegral”.

In[40]:=
```
X = (1/9 Gamma[1/3]^2 Zeta[5/3])^(3/10);
Y = -Sqrt[Pi] Zeta[1/2] Zeta[3/2];

A₁ = 5 X^2;
A₂ = X^-1 Y;
A₃ = 3/80 X^-4 Y^2;
A₄ = 11/3200 X^-7 Y^3;

IntriguingIntegral =
  NIntegrate[1/2 × (Zeta[3/2 + ⅈ t] Zeta[-3/2 - ⅈ t])/((3/2 + ⅈ t) Cosh[π t]), {t, -Infinity, Infinity},
   WorkingPrecision → 25, PrecisionGoal → 20, AccuracyGoal → 20];
WittenZetaDerivativeAt0 = 1/12 (1 + EulerGamma) + 3/4 Log[2 Pi] -
   2 Zeta'[-1] + IntriguingIntegral;
WittenZetaDerivativeAt0SecondFormula = Log[2 π]; (* A much simplified
 formula proved in a 2016 paper by J. Borwein and K. Dilcher *)

K = Sqrt[3]/Sqrt[5 π] X^(1/3) Exp[-1/2560 X^-10 Y^4 + WittenZetaDerivativeAt0];

Ksecondformula = (2 Sqrt[3 π])/Sqrt[5] X^(1/3) Exp[-1/2560 X^-10 Y^4];
```

Now let's print the numerical values of A_1, A_2, A_3, A_4.

In[51]:=
```
N[{A₁, A₂, A₃, A₄}, 20]
```

Out[51]=
```
{6.8582604761631263709, 5.7736017451051146318,
 0.91134107257243679797, 0.35163754155820937624}
```

Run the next instruction to get the numerical value of the "intriguing integral" that appears in our formula for $\omega'(0)$.

In[52]:=
```
IntriguingIntegral
```

Out[52]=
```
-0.002807659540359892588217126
```

The next instruction evaluates ω'(0).

In[53]:=
```
WittenZetaDerivativeAt0
```
Out[53]=
```
1.837877066409345483560659473
```

Evaluate the next command to verify that the Borwein-Dilcher formula gives numerically the same result as our formula.

In[54]:=
```
N[WittenZetaDerivativeAt0SecondFormula, 28]
```
Out[54]=
```
1.837877066409345483560659473
```

The next two instructions output the value of the constant K, computed using each of the two equivalent formulas in the paper.

In[55]:=
```
K
```
Out[55]=
```
2.446290334864178943325432671
```

In[56]:=
```
N[Ksecondformula, 28]
```
Out[56]=
```
2.446290334864178943325432671
```

4. Simplification of constants

In this section we verify the simplification of the constants B_1, B_2, B_3, B_4, B_5 that appears in section 11 of the paper. Run the code below to define the relevant constants μ_1, μ_2, τ_1, τ_2, τ_3, u, v, B_1, B_2, B_3, B_4, B_5.

In[57]:=

$$\mu_1 = \frac{2^{2/3}}{3} \text{Gamma}\left[\frac{1}{3}\right]^2 \text{Zeta}\left[\frac{5}{3}\right];$$

$$\mu_2 = \sqrt{2\pi}\ \text{Zeta}\left[\frac{1}{2}\right] \text{Zeta}\left[\frac{3}{2}\right];$$

$$\tau_1 = 2\, X^2;$$

$$\tau_2 = \frac{3}{10} X^{-1}\, Y;$$

$$\tau_3 = \frac{3}{400} X^{-4}\, Y^2;$$

$$u = \tau_2 / \tau_1;$$

$$v = \tau_3 / \tau_1;$$

$$B_1 = \mu_1\, \tau_1{}^{-2/3};$$

$$B_2 = \frac{2}{3} \mu_1\, \tau_1{}^{-2/3}\, u + \mu_2\, \tau_1{}^{-1/2};$$

$$B_3 = \frac{2}{3} \mu_1\, \tau_1{}^{-2/3}\, v + \frac{5}{9} \mu_1\, \tau_1{}^{-2/3}\, u^2 + \frac{1}{2} \mu_2\, \tau_1{}^{-1/2}\, u;$$

$$B_4 = \frac{10}{9} \mu_1\, \tau_1{}^{-2/3}\, u\, v + \frac{40}{81} \mu_1\, \tau_1{}^{-2/3}\, u^3 + \frac{1}{2} \mu_2\, \tau_1{}^{-1/2}\, v + \frac{3}{8} \mu_2\, \tau_1{}^{-1/2}\, u^2;$$

$$B_5 = \frac{5}{9} \mu_1\, \tau_1{}^{-2/3}\, v^2 + \frac{40}{27} \mu_1\, \tau_1{}^{-2/3}\, u^2\, v + \frac{110}{243} \mu_1\, \tau_1{}^{-2/3}\, u^4 + \frac{3}{4} \mu_2\, \tau_1{}^{-1/2}\, u\, v + \frac{5}{16} \mu_2\, \tau_1{}^{-1/2}\, u^3;$$

In the paper it is claimed that each of B_1, B_2, B_3, B_4, B_5 is a multiple of a certain power of X times a certain power of Y by some explicit rational numbers. The next line of code computes these numbers:

In[69]:=

$$\{B_1 / (X^2),\ B_2 / (X^{-1}\, Y),\ B_3 / (X^{-4}\, Y^2),\ B_4 / (X^{-7}\, Y^3),\ B_5 / (X^{-10}\, Y^4)\} \text{ // Simplify}$$

Out[69]=

$$\left\{3,\ -\frac{7}{10},\ -\frac{3}{100},\ -\frac{11}{3200},\ -\frac{1}{2560}\right\}$$

5. Numerical computation of the sequence r(n)

In this section we compute the initial values of the sequence r(n) defined section 1.1 of the paper. This is done by expanding the generating function given in equation (2) as a power series. Run the code below to define the relevant functions.

In[70]:=
```
dimtable[a_] := Table[1 / 2 m n (m + n), {m, 1, a}, {n, 1, a}];
irrepdims[bound_] :=
  Sort[Select[Flatten[Table[1 / 2 m n (m + n), {m, 1, Ceiling[Sqrt[2 bound]]},
      {n, 1, Ceiling[Sqrt[2 bound]]}]], # ≤ bound &]];
rvalues[n_] := Module[{dims},
   dims = irrepdims[n];
   CoefficientList[
    Series[Product[1 / (1 - x^dims[[j]]), {j, 1, Length[dims]}], {x, 0, n}], x]
  ];
```

Calling the function rvalues[n] will now compute the values r(k) for k=0,1,...,n. Try running the line below.

In[73]:=
```
rvalues[100]
```

Out[73]=
```
{1, 1, 1, 3, 3, 3, 8, 8, 9, 17, 19, 21, 35, 39, 44, 68, 79, 87, 127, 145, 162,
 228, 261, 291, 395, 451, 506, 665, 760, 850, 1096, 1254, 1400, 1765, 2016,
 2249, 2800, 3188, 3556, 4356, 4953, 5522, 6688, 7581, 8447, 10 123, 11 464,
 12 747, 15 141, 17 094, 18 997, 22 395, 25 235, 27 998, 32 766, 36 835, 40 846,
 47 484, 53 281, 58 997, 68 200, 76 381, 84 479, 97 130, 108 569, 119 933, 137 295,
 153 148, 168 986, 192 576, 214 437, 236 363, 268 318, 298 225, 328 350, 371 348,
 412 114, 453 197, 510 842, 565 952, 621 727, 698 651, 772 826, 848 021, 950 191,
 1 049 424, 1 150 455, 1 285 552, 1 417 749, 1 552 557, 1 730 541, 1 905 830, 2 084 988,
 2 318 476, 2 549 868, 2 786 738, 3 092 136, 3 396 181, 3 708 118, 4 105 813, 4 503 868}
```